// File: /Users/gge/ribbon-finance/avocado/contracts/core/Rewards.sol



/**
 * SPDX-License-Identifier: UNLICENSED
 */
pragma solidity =0.6.10;

import {RewardsInterface} from "../interfaces/RewardsInterface.sol";
import {ERC20Interface} from "../interfaces/ERC20Interface.sol";
import {OtokenInterface} from "../interfaces/OtokenInterface.sol";
import {AddressBookInterface} from "../interfaces/AddressBookInterface.sol";
import {WhitelistInterface} from "../interfaces/WhitelistInterface.sol";
import {SafeMath} from "../packages/oz/SafeMath.sol";
import {SafeERC20} from "../packages/oz/SafeERC20.sol";
import {Ownable} from "../packages/oz/Ownable.sol";

/**
 * @title Rewards
 * @author 10 Delta
 * @notice The rewards module distributes liquidity mining rewards to oToken shorters
 */
contract Rewards is Ownable, RewardsInterface {
    using SafeMath for uint256;
    using SafeERC20 for ERC20Interface;

    /// @notice AddressBook module address
    AddressBookInterface public immutable addressBook;
    /// @notice token to pay rewards in
    ERC20Interface public rewardsToken;
    /// @notice address for notifying/removing rewards
    address public rewardsDistribution;
    /// @notice if true all vault owners will receive rewards
    bool public whitelistAllOwners;
    /// @notice if true rewards get forfeited when closing an oToken before expiry
    bool public forfeitRewards;
    /// @notice total rewards forfeited
    uint256 public forfeitedRewards;
    /// @notice the reward rate for an oToken
    mapping(address => uint256) public rewardRate;
    /// @notice the last upddate time for an oToken
    mapping(address => uint256) public lastUpdateTime;
    /// @notice the reward per token for an oToken
    mapping(address => uint256) public rewardPerTokenStored;
    /// @notice the reward per token for a user
    mapping(address => mapping(address => uint256)) public userRewardPerTokenPaid;
    /// @notice the rewards earnt by a user
    mapping(address => mapping(address => uint256)) public rewards;
    /// @notice the total amount of each oToken
    mapping(address => uint256) public totalSupply;
    /// @notice the balances for each oToken
    mapping(address => mapping(address => uint256)) public balances;

    /**
     * @dev constructor
     * @param _addressBook AddressBook module address
     */
    constructor(address _addressBook) public {
        require(_addressBook != address(0), "Rewards: Invalid address book");

        addressBook = AddressBookInterface(_addressBook);
    }

    /// @notice emits an event when the whitelist all owners setting changes
    event WhitelistAllOwners(bool whitelisted);
    /// @notice emits an event when the forfeit rewards setting changes
    event ForfeitRewards(bool forfeited);
    /// @notice emits an event the rewards distribution address is updated
    event RewardsDistributionUpdated(address indexed _rewardsDistribution);
    /// @notice emits an event when rewards are added for an otoken
    event RewardAdded(address indexed otoken, uint256 reward);
    /// @notice emits an event when rewards are removed for an otoken
    event RewardRemoved(address indexed token, uint256 amount);
    /// @notice emits an event when forfeited rewards are recovered
    event RecoveredForfeitedRewards(uint256 amount);
    /// @notice emits an event when rewards are paid out to a user
    event RewardPaid(address indexed otoken, address indexed account, uint256 reward);

    /**
     * @notice check if the sender is the controller module
     */
    modifier onlyController() {
        require(msg.sender == addressBook.getController(), "Rewards: Sender is not Controller");

        _;
    }

    /**
     * @notice check if the sender is the rewards distribution address
     */
    modifier onlyRewardsDistribution() {
        require(msg.sender == rewardsDistribution, "Rewards: Sender is not RewardsDistribution");

        _;
    }

    /**
     * @notice initializes the rewards token
     * @dev can only be called by the owner, doesn't allow reinitializing the rewards token
     * @param _rewardsToken the address of the rewards token
     */
    function setRewardsToken(address _rewardsToken) external onlyOwner {
        require(rewardsToken == ERC20Interface(0), "Rewards: Token already set");

        rewardsToken = ERC20Interface(_rewardsToken);
    }

    /**
     * @notice sets the rewards distribution address
     * @dev can only be called by the owner
     * @param _rewardsDistribution the new rewards distribution address
     */
    function setRewardsDistribution(address _rewardsDistribution) external onlyOwner {
        rewardsDistribution = _rewardsDistribution;

        emit RewardsDistributionUpdated(_rewardsDistribution);
    }

    /**
     * @notice whitelists/blacklists all vault owners from receiving rewards
     * @dev can only be called by the owner
     * @param _whitelistAllOwners new boolean value to set whitelistAllOwners to
     */
    function setWhitelistAllOwners(bool _whitelistAllOwners) external onlyOwner {
        whitelistAllOwners = _whitelistAllOwners;

        emit WhitelistAllOwners(_whitelistAllOwners);
    }

    /**
     * @notice enables/disables forfeiting of rewards when closing an oToken short before expiry
     * @dev can only be called by the owner
     * @param _forfeitRewards new boolean value to set forfeitRewards to
     */
    function setForfeitRewards(bool _forfeitRewards) external onlyOwner {
        forfeitRewards = _forfeitRewards;

        emit ForfeitRewards(_forfeitRewards);
    }

    /**
     * @notice adds rewards for an oToken
     * @dev can only be called by rewards distribution, reverts if rewardsToken is uninitialized
     * @param otoken oToken to add rewards for
     * @param reward amount of rewards to add
     */
    function notifyRewardAmount(address otoken, uint256 reward) external onlyRewardsDistribution {
        _updateReward(otoken, address(0));

        // Reverts if oToken is expired
        uint256 remaining = OtokenInterface(otoken).expiryTimestamp().sub(now);
        uint256 _rewardRate = (reward.div(remaining)).add(rewardRate[otoken]);

        // Transfer reward amount to this contract
        // This is necessary because there can be multiple rewards so a balance check isn't enough
        rewardsToken.safeTransferFrom(msg.sender, address(this), reward);

        rewardRate[otoken] = _rewardRate;
        lastUpdateTime[otoken] = now;

        emit RewardAdded(otoken, reward);
    }

    /**
     * @notice remove rewards for an oToken
     * @dev can only be called by rewards distribution, reverts if rewardsToken is uninitialized
     * @param otoken oToken to remove rewards from
     * @param rate rewardRate to remove for an oToken
     */
    function removeRewardAmount(address otoken, uint256 rate) external onlyRewardsDistribution {
        _updateReward(otoken, address(0));

        // Reverts if oToken is expired
        uint256 _rewardRate = rewardRate[otoken];
        uint256 reward = (OtokenInterface(otoken).expiryTimestamp().sub(now)).mul(rate);

        // Transfers the removed reward amount back to the owner
        rewardsToken.safeTransfer(msg.sender, reward);

        rewardRate[otoken] = _rewardRate.sub(rate);
        lastUpdateTime[otoken] = now;

        emit RewardRemoved(otoken, reward);
    }

    /**
     * @notice transfers forfeited rewards to rewards distribution
     * @dev can only be called by rewards distribution
     */
    function recoverForfeitedRewards() external onlyRewardsDistribution {
        uint256 _forfeitedRewards = forfeitedRewards;
        ERC20Interface _rewardsToken = rewardsToken;
        uint256 rewardsBalance = _rewardsToken.balanceOf(address(this));
        _rewardsToken.safeTransfer(msg.sender, _forfeitedRewards < rewardsBalance ? _forfeitedRewards : rewardsBalance);
        forfeitedRewards = 0;
        emit RecoveredForfeitedRewards(_forfeitedRewards);
    }

    /**
     * @notice records the minting of oTokens
     * @dev can only be called by the controller
     * @param otoken oToken address
     * @param account vault owner
     * @param amount amount of oTokens minted
     */
    function mint(
        address otoken,
        address account,
        uint256 amount
    ) external override onlyController {
        if (isWhitelistedOwner(account)) {
            _updateReward(otoken, account);
            totalSupply[otoken] = totalSupply[otoken].add(amount);
            balances[otoken][account] = balances[otoken][account].add(amount);
        }
    }

    /**
     * @notice records the burning of oTokens
     * @dev can only be called by the controller
     * @param otoken oToken address
     * @param account vault owner
     * @param amount amount of oTokens burnt
     */
    function burn(
        address otoken,
        address account,
        uint256 amount
    ) external override onlyController {
        uint256 balance = balances[otoken][account];
        if (balance > 0) {
            _updateReward(otoken, account);
            amount = amount > balance ? balance : amount;
            totalSupply[otoken] = totalSupply[otoken].sub(amount);
            balances[otoken][account] = balance.sub(amount);
            if (forfeitRewards) {
                uint256 _rewards = rewards[otoken][account];
                uint256 _forfeitedRewards = _rewards.mul(amount).div(balance);
                rewards[otoken][account] = _rewards.sub(_forfeitedRewards);
                forfeitedRewards = forfeitedRewards.add(_forfeitedRewards);
            }
        }
    }

    /**
     * @notice claims oToken minting rewards and transfers it to the vault owner
     * @dev can only be called by the controller
     * @param otoken oToken address
     * @param account vault owner
     */
    function getReward(address otoken, address account) external override onlyController {
        _getReward(otoken, account);
    }

    /**
     * @notice allows vault owners to claim rewards if this module has been deprecated
     * @dev oToken must be expired before claiming rewards
     * @param otoken oToken address
     */
    function claimReward(address otoken) external {
        require(address(this) != addressBook.getRewards(), "Rewards: Module not deprecated");
        require(now >= OtokenInterface(otoken).expiryTimestamp(), "Rewards: Not expired");
        _getReward(otoken, msg.sender);
    }

    /**
     * @notice checks if a vault owner is whitelisted
     * @param account vault owner
     * @return boolean, True if the vault owner is whitelisted
     */
    function isWhitelistedOwner(address account) public view returns (bool) {
        return whitelistAllOwners || WhitelistInterface(addressBook.getWhitelist()).isWhitelistedOwner(account);
    }

    /**
     * @notice returns the last time rewards are applicable for an oToken
     * @param otoken oToken address
     * @return last time rewards are applicable for the oToken
     */
    function lastTimeRewardApplicable(address otoken) public view returns (uint256) {
        uint256 periodFinish = OtokenInterface(otoken).expiryTimestamp();
        return periodFinish > now ? now : periodFinish;
    }

    /**
     * @notice returns the reward per token for an oToken
     * @param otoken oToken address
     * @return reward per token for the oToken
     */
    function rewardPerToken(address otoken) public view returns (uint256) {
        uint256 _totalSupply = totalSupply[otoken];
        if (_totalSupply == 0) {
            return rewardPerTokenStored[otoken];
        }
        return
            rewardPerTokenStored[otoken].add(
                lastTimeRewardApplicable(otoken).sub(lastUpdateTime[otoken]).mul(rewardRate[otoken]).mul(1e18).div(
                    _totalSupply
                )
            );
    }

    /**
     * @notice returns the rewards a vault owner has earnt for an oToken
     * @param otoken oToken address
     * @param account vault owner address
     * @return rewards earnt by the vault owner
     */
    function earned(address otoken, address account) public view returns (uint256) {
        return
            balances[otoken][account]
                .mul(rewardPerToken(otoken).sub(userRewardPerTokenPaid[otoken][account]))
                .div(1e18)
                .add(rewards[otoken][account]);
    }

    /**
     * @dev updates the reward per token and the rewards earnt by a vault owner
     * @param otoken oToken address
     * @param account vault owner address
     */
    function _updateReward(address otoken, address account) internal {
        uint256 _rewardPerTokenStored = rewardPerToken(otoken);
        rewardPerTokenStored[otoken] = _rewardPerTokenStored;
        if (account != address(0)) {
            lastUpdateTime[otoken] = lastTimeRewardApplicable(otoken);
            rewards[otoken][account] = earned(otoken, account);
            userRewardPerTokenPaid[otoken][account] = _rewardPerTokenStored;
        }
    }

    /**
     * @dev claims oToken minting rewards and transfers it to the vault owner
     * @param otoken oToken address
     * @param account vault owner
     */
    function _getReward(address otoken, address account) internal {
        _updateReward(otoken, account);
        uint256 reward = rewards[otoken][account];
        if (reward > 0) {
            rewards[otoken][account] = 0;
            ERC20Interface _rewardsToken = rewardsToken;
            uint256 rewardsBalance = _rewardsToken.balanceOf(address(this));
            if (rewardsBalance > 0) {
                _rewardsToken.safeTransfer(account, reward < rewardsBalance ? reward : rewardsBalance);
                emit RewardPaid(otoken, account, reward);
            }
        }
    }
}

// File: /Users/gge/ribbon-finance/avocado/contracts/interfaces/AddressBookInterface.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity 0.6.10;

interface AddressBookInterface {
    /* Getters */

    function getOtokenImpl() external view returns (address);

    function getOtokenFactory() external view returns (address);

    function getWhitelist() external view returns (address);

    function getController() external view returns (address);

    function getOracle() external view returns (address);

    function getRewards() external view returns (address);

    function getMarginPool() external view returns (address);

    function getMarginCalculator() external view returns (address);

    function getLiquidationManager() external view returns (address);

    function getAddress(bytes32 _id) external view returns (address);

    /* Setters */

    function setOtokenImpl(address _otokenImpl) external;

    function setOtokenFactory(address _factory) external;

    function setWhitelist(address _whitelist) external;

    function setController(address _controller) external;

    function setMarginPool(address _marginPool) external;

    function setMarginCalculator(address _calculator) external;

    function setLiquidationManager(address _liquidationManager) external;

    function setOracle(address _oracle) external;

    function setRewards(address _rewards) external;

    function setAddress(bytes32 _id, address _newImpl) external;
}

// File: /Users/gge/ribbon-finance/avocado/contracts/interfaces/ERC20Interface.sol



/**
 * SPDX-License-Identifier: UNLICENSED
 */
pragma solidity 0.6.10;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface ERC20Interface {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address sender,
        address recipient,
        uint256 amount
    ) external returns (bool);

    function decimals() external view returns (uint8);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// File: /Users/gge/ribbon-finance/avocado/contracts/interfaces/OtokenInterface.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity 0.6.10;

interface OtokenInterface {
    function underlyingAsset() external view returns (address);

    function strikeAsset() external view returns (address);

    function collateralAsset() external view returns (address);

    function strikePrice() external view returns (uint256);

    function expiryTimestamp() external view returns (uint256);

    function isPut() external view returns (bool);

    function isWhitelisted() external view returns (bool);

    function init(
        address _addressBook,
        address _underlyingAsset,
        address _strikeAsset,
        address _collateralAsset,
        uint256 _strikePrice,
        uint256 _expiry,
        bool _isPut,
        bool _isWhitelisted
    ) external;

    function getOtokenDetails()
        external
        view
        returns (
            address,
            address,
            address,
            uint256,
            uint256,
            bool
        );

    function mintOtoken(address account, uint256 amount) external;

    function burnOtoken(address account, uint256 amount) external;
}

// File: /Users/gge/ribbon-finance/avocado/contracts/interfaces/RewardsInterface.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity 0.6.10;

interface RewardsInterface {
    function mint(
        address otoken,
        address account,
        uint256 amount
    ) external;

    function burn(
        address otoken,
        address account,
        uint256 amount
    ) external;

    function getReward(address otoken, address account) external;
}

// File: /Users/gge/ribbon-finance/avocado/contracts/interfaces/WhitelistInterface.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity 0.6.10;

interface WhitelistInterface {
    /* View functions */

    function addressBook() external view returns (address);

    function isWhitelistedProduct(
        address _underlying,
        address _strike,
        address _collateral,
        bool _isPut
    ) external view returns (bool);

    function isOwnerWhitelistedProduct(
        address _underlying,
        address _strike,
        address _collateral,
        bool _isPut
    ) external view returns (bool);

    function isWhitelistedCollateral(address _collateral) external view returns (bool);

    function isWhitelistedOtoken(address _otoken) external view returns (bool);

    function isOwnerWhitelistedOtoken(address _otoken) external view returns (bool);

    function isWhitelistedCallee(address _callee) external view returns (bool);

    function isWhitelistedOwner(address _owner) external view returns (bool);

    function whitelistProduct(
        address _underlying,
        address _strike,
        address _collateral,
        bool _isPut
    ) external;

    function whitelistCollateral(address _collateral) external;

    /* Admin / factory only functions */
    function ownerWhitelistProduct(
        address _underlying,
        address _strike,
        address _collateral,
        bool _isPut
    ) external;

    function blacklistProduct(
        address _underlying,
        address _strike,
        address _collateral,
        bool _isPut
    ) external;

    function ownerWhitelistCollateral(address _collateral) external;

    function blacklistCollateral(address _collateral) external;

    function whitelistOtoken(address _otoken) external;

    function blacklistOtoken(address _otoken) external;

    function whitelistCallee(address _callee) external;

    function blacklistCallee(address _callee) external;

    function whitelistOwner(address account) external;

    function blacklistOwner(address account) external;
}

// File: /Users/gge/ribbon-finance/avocado/contracts/packages/oz/Address.sol



// SPDX-License-Identifier: MIT
// openzeppelin-contracts v3.1.0

pragma solidity ^0.6.2;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        // solhint-disable-next-line no-inline-assembly
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        // solhint-disable-next-line avoid-low-level-calls, avoid-call-value
        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain`call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return _functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        return _functionCallWithValue(target, data, value, errorMessage);
    }

    function _functionCallWithValue(
        address target,
        bytes memory data,
        uint256 weiValue,
        string memory errorMessage
    ) private returns (bytes memory) {
        require(isContract(target), "Address: call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.call{value: weiValue}(data);
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                // solhint-disable-next-line no-inline-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: /Users/gge/ribbon-finance/avocado/contracts/packages/oz/Context.sol



// SPDX-License-Identifier: MIT
// openzeppelin-contracts v3.1.0

pragma solidity 0.6.10;

/*
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with GSN meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address payable) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes memory) {
        this; // silence state mutability warning without generating bytecode - see https://github.com/ethereum/solidity/issues/2691
        return msg.data;
    }
}

// File: /Users/gge/ribbon-finance/avocado/contracts/packages/oz/Ownable.sol



// SPDX-License-Identifier: MIT
// openzeppelin-contracts v3.1.0

pragma solidity 0.6.10;

import "./Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() internal {
        address msgSender = _msgSender();
        _owner = msgSender;
        emit OwnershipTransferred(address(0), msgSender);
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(_owner == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        emit OwnershipTransferred(_owner, address(0));
        _owner = address(0);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        emit OwnershipTransferred(_owner, newOwner);
        _owner = newOwner;
    }
}

// File: /Users/gge/ribbon-finance/avocado/contracts/packages/oz/SafeERC20.sol



// SPDX-License-Identifier: MIT
// openzeppelin-contracts v3.1.0

/* solhint-disable */
pragma solidity ^0.6.0;

import "../../interfaces/ERC20Interface.sol";
import "./SafeMath.sol";
import "./Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for ERC20Interface;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using SafeMath for uint256;
    using Address for address;

    function safeTransfer(
        ERC20Interface token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        ERC20Interface token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {ERC20Interface-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        ERC20Interface token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        // solhint-disable-next-line max-line-length
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        ERC20Interface token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender).add(value);
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        ERC20Interface token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender).sub(
            value,
            "SafeERC20: decreased allowance below zero"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(ERC20Interface token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            // solhint-disable-next-line max-line-length
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: /Users/gge/ribbon-finance/avocado/contracts/packages/oz/SafeMath.sol



// SPDX-License-Identifier: MIT
// openzeppelin-contracts v3.1.0

/* solhint-disable */
pragma solidity ^0.6.0;

/**
 * @dev Wrappers over Solidity's arithmetic operations with added overflow
 * checks.
 *
 * Arithmetic operations in Solidity wrap on overflow. This can easily result
 * in bugs, because programmers usually assume that an overflow raises an
 * error, which is the standard behavior in high level programming languages.
 * `SafeMath` restores this intuition by reverting the transaction when an
 * operation overflows.
 *
 * Using this library instead of the unchecked operations eliminates an entire
 * class of bugs, so it's recommended to use it always.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        uint256 c = a + b;
        require(c >= a, "SafeMath: addition overflow");

        return c;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        return sub(a, b, "SafeMath: subtraction overflow");
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     * - Subtraction cannot overflow.
     */
    function sub(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        require(b <= a, errorMessage);
        uint256 c = a - b;

        return c;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
        // benefit is lost if 'b' is also tested.
        // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
        if (a == 0) {
            return 0;
        }

        uint256 c = a * b;
        require(c / a == b, "SafeMath: multiplication overflow");

        return c;
    }

    /**
     * @dev Returns the integer division of two unsigned integers. Reverts on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        return div(a, b, "SafeMath: division by zero");
    }

    /**
     * @dev Returns the integer division of two unsigned integers. Reverts with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     * - The divisor cannot be zero.
     */
    function div(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        // Solidity only automatically asserts when dividing by 0
        require(b > 0, errorMessage);
        uint256 c = a / b;
        // assert(a == b * c + a % b); // There is no case in which this doesn't hold

        return c;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * Reverts when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        return mod(a, b, "SafeMath: modulo by zero");
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * Reverts with custom message when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     * - The divisor cannot be zero.
     */
    function mod(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        require(b != 0, errorMessage);
        return a % b;
    }
}